Bifurcation diagram for lower order periods of the logisitc map

define map

```
> restart;
> f := x-> a*x*(1-x);
```

$$f := x \to a\,x\,(1-x) \quad (1)$$

period 1 and 2

```
> s2:=solve(f(f(x))=x,x);
```

$$s2 := 0,\ \frac{-1+a}{a},\ \frac{\frac{1}{2}\,a+\frac{1}{2}+\frac{1}{2}\sqrt{-3-2\,a+a^2}}{a},\ \frac{\frac{1}{2}\,a+\frac{1}{2}-\frac{1}{2}\sqrt{-3-2\,a+a^2}}{a} \quad (2)$$

```
> x1:=s2[1];
  x2:=s2[2];
  x21:=s2[3];
  x22:=s2[4];
```

$$x1 := 0$$
$$x2 := \frac{-1+a}{a}$$
$$x21 := \frac{\frac{1}{2}\,a+\frac{1}{2}+\frac{1}{2}\sqrt{-3-2\,a+a^2}}{a}$$
$$x22 := \frac{\frac{1}{2}\,a+\frac{1}{2}-\frac{1}{2}\sqrt{-3-2\,a+a^2}}{a} \quad (3)$$

derivative of the map

```
> fp := x-> a*(1-x)-a*x;
```

$$fp := x \to a\,(1-x)-a\,x \quad (4)$$

Multiplier for period 2 orbit

```
> M2:=expand(fp(x21)*fp(x22));
```

$$M2 := 4+2\,a-a^2 \quad (5)$$

Intersections with the stability region

```
> solve(M2=1);
  solve(M2=-1);
```

$$-1,\ 3$$
$$1-\sqrt{6},\ 1+\sqrt{6} \quad (6)$$

```
> evalf(%);
```

$$-1.449489743,\ 3.449489743 \quad (7)$$

Plot stability region and multiplier

```
> plot({M2,1,-1},a=-2..4,thickness=5,font=[COURIER,BOLD,20]);
```



Plot p1 and p2 orbits

```
> with(plots):
> pl1:=plot({x1,x2},a=0.9..4,thickness=5,font=[COURIER,BOLD,20]):
> pl2:=plot({x21,x22},a=0.9..4,thickness=5,font=[COURIER,BOLD,20]
  ):
  display([pl1,pl2],thickness=5,font=[COURIER,BOLD,20]);
```



period 4

```
> p4:=(solve(f(f(f(f(x))))=x,x));
```

$$p4 := 0,\ \frac{-1+a}{a},\ \frac{\frac{1}{2}\,a+\frac{1}{2}+\frac{1}{2}\sqrt{-3-2\,a+a^2}}{a},\ \frac{\frac{1}{2}\,a+\frac{1}{2}-\frac{1}{2}\sqrt{-3-2\,a+a^2}}{a},\ \frac{1}{a}\big(RootOf(1+a^2+(-a^4-a^3-a^2-a)\,\_Z+(2\,a^5+a^4+4\,a^3+a^2+2\,a)\,\_Z^2+(-1-4\,a^5-a^6-5\,a^2-4\,a^3-5\,a^4)\,\_Z^3+(3\,a^6+5\,a^5+14\,a^4+4\,a^3+6\,a^2+2\,a)\,\_Z^4+(-3\,a^6-12\,a^5-12\,a^4-18\,a^3-a^2-4\,a)\,\_Z^5+(a^6+10\,a^2+18\,a^4+1+15\,a^5+17\,a^3)\,\_Z^6+(-6\,a^5-30\,a^4-12\,a^3-14\,a^2-2\,a)\,\_Z^7+(3\,a^2+15\,a^4+6\,a+30\,a^3)\,\_Z^8+(-15\,a^2-1-20\,a^3)\,\_Z^9+(15\,a^2+3\,a)\,\_Z^{10}-6\,a\,\_Z^{11}+\_Z^{12})\big) \quad (8)$$

```
> p40:=allvalues(p4[5]):;
> #plot({p40[1],p40[2],p40[3],p40[4],p40[5],p40[6],p40[7],p40[8],
  p40[9],p40[10],p40[11],p40[12]},a=1+sqrt(6)..3.96,thickness=5,
  font=[COURIER,BOLD,20]):
> pl4:=plot({p40[1],p40[2],p40[3],p40[4]},a=1+sqrt(6)+1e-3..3.96,
  thickness=5,font=[COURIER,BOLD,30],numpoints=2):
  display([pl4],thickness=5,font=[COURIER,BOLD,20]);
```



all p1, p2 and p4 orbits:

```
> display([pl1,pl2,pl4],thickness=5,font=[COURIER,BOLD,20]);
```



period 3

```
> p3:=(solve(f(f(f(x)))=x,x));
```

(9)

$$p3 := 0,\ \frac{-1+a}{a},\ \frac{1}{a}\left(RootOf\left(1+a+a^2+\left(-2\,a^2-1-2\,a-a^3\right)\_Z+\left(3\,a+1+2\,a^3+3\,a^2\right)\_Z^2+\left(-3\,a-1-5\,a^2-a^3\right)\_Z^3+\left(4\,a+1+3\,a^2\right)\_Z^4+\left(-1-3\,a\right)\_Z^5+\_Z^6\right)\right) \quad (9)$$

```
> p30:=allvalues(p3[3]):
> pl3:=plot({p30[1],p30[2],p30[3],p30[4],p30[5],p30[6]},a=3.8..4,
  thickness=5,font=[COURIER,BOLD,20],numpoints=10):
  display([pl3],thickness=5,font=[COURIER,BOLD,20]);
```



all p1, p2, p3 and p4 orbits:

```
> display([pl1,pl2,pl3,pl4],thickness=5,font=[COURIER,BOLD,20]);
```